
KALMAN (BPRIOR = 事前ベクトル,
 BTRANS = 遷移方程式の係数行列,
 EMEAS, ETRANS, PRINT, SILENT, SMOOTH, TERSE,
 VBPRIOR = 事前分布の分散, VMEAS = 観測方程式の分散要素,
 VTRANS = 遷移方程式の分散要素, XFIXED = 観測方程式の X 行列)
 従属変数リスト [| 独立変数のリスト] ;

機能 :

KALMAN は線形モデルをカルマンフィルタ法で推定します。この方法は、係数が時間を通じて確率過程に従う回帰タイプのモデルを推定するのに通常は使われますが、かなり一般的な状態空間モデルを扱うことができます。

使用法:

カルマンフィルタモデルは、次の2つの部分からできています (状態空間型)。

$$\begin{aligned}
 y_t &= X_t \beta_t + \varepsilon_t & \varepsilon_t &\sim N(0, \sigma^2 H) & (\text{観測方程式}) \\
 & \quad (n \times 1) & \quad (n \times m)(m \times 1) & \quad (n \times n) \\
 \beta_t &= T^* \beta_{t-1} + \eta_t & \eta_t &\sim N(0, \sigma^2 Q) & (\text{遷移方程式}) \\
 & \quad (m \times m) & \quad (m \times m) & \quad (m \times m) \\
 \beta_0 &\sim N(\beta_0, \sigma^2 P_0) & & & (\text{初期条件})
 \end{aligned}$$

行列 T , H , Q は既知であると仮定されており、KALMAN オプションリストでユーザが与えていない限り、単位行列をそれぞれ既定値とします。これらは時間で変化することは想定していませんが、標本期間に渡ってのループ内で KALMAN を走らせることによって簡単に緩めることができます。NOEMEAS と NOETRANS オプションは、観測方程式と遷移方程式それぞれの分散をゼロにするのに用いられます。

y_t と X_t 変数は、ちょうど最小2乗法の場合のように従属変数と独立変数になります。従属変数を1つ以上用いたい場合は、最初にすべての y 変数、次に |、そしてそれぞれの y に対して X 変数をリストします (すべての y に対して同じなら X リストを重ねます)。方程式間の制約を防ぐために X 変数とともにゼロを含めることができます (例を参照)。 X_t が標本で固定されているなら、XFIXED オプションを用います。

時変パラメータモデルを得るには、VTRANS=Q(ノイズ信号比)とBTRANS=T(単位行列でない場合)とします。

一般の ARMA(p, q) モデルに対する尤度関数を評価するには、XFIXED 行列をモデルの推定された係数で埋めます。この機能と理由についての詳しい説明は Harvey, p.103 を参照して下さい。

オプション:

BPRIOR = 観測方程式の事前係数 β_0 のベクトル。XFIXED を用いている場合は必須です。その他の場合は、標本の初期観測値での回帰から既定値として計算されます。最初の M 個の観測値だけでは事前情報を識別するのに十分でない場合は、1つの観測値が加えられ、BPRIOR が再推定されます。

BTRANS = T = 遷移方程式の係数行列 (既定値は単位行列)

EMEAS/NOEMEAS は観測方程式に誤差項があることを示しています (NOEMEAS あるいは NOEM は VMEAS=ゼロと同じ).

ETRANS/NOETRANS は遷移方程式に誤差項があることを示しています.

PRINT/NOPRINT は事前分布, @STATE, @RES1, @RECRES, @SMOOTH, @RES2 をプリントします.

SILENT/NOSILENT はほとんどのアウトプットを出さないようにします.

SMOOTH/NOSMOOTH 状態ベクトル β_t の平滑化推定値 (@SMOOTH で保存) と直接の残差 @RES2 を計算します.

VBPRIOR = P_0 = 事前分布 (対称行列) の分散. BPRIOR を指定していれば必須です. σ^2 は, この行列から分解されることに注意して下さい.

VMEAS = H = 観測方程式の分散要素 (対称行列). 既定値は単位行列です. Harvey では SHS' と表しています.

VTRANS = Q = 遷移方程式の分散要素 (対称行列). 既定値は単位行列です. H=単位行列なら ノイズ信号比を指定します. Harvey では RQR' と表しています.

XFIXED = X 時間を通じて固定されているときの観測方程式の X 行列です.

例:

最も簡単なカルマンフィルタモデルは OLSQ と同じです ($\beta_t = \beta_{t-1} = \beta$ という遷移方程式を用いている). これは次のコマンドで推定できます.

```
KALMAN(NOETRANS) Y C X;
```

これは OLSQ Y C X; と同じ係数推定値を得ますが, 逐次残差と共に逐次的に計算します.

β_t が非対角分散行列のランダム・ウォークに従う Cooley-Prescott の適応期待モデルを推定するには, 次のようにします.

```
KALMAN(VTRANS=NSRMAT) Y C X1 X2;
```

確率的に収束するパラメータのモデル (この場合, 遷移行列が 1 より小さい根を持つのでゼロに収束)

```
MFORM(TYPE=DIAG,NROW=3) TMAT=.9;
KALMAN(BTRANS=TMAT,VTRANS=NSRMAT) Y C X1 X2;
```

従属変数が 2 つの場合; 外生変数の 2 つのリストは必ず同じ長さであることに注意して下さい. この場合, 両方程式は必ず同じ 2 つの係数をもつようになります.

```
KALMAN(NOET) Y1 Y2 | C1 X1, C2 X2;
```

従属変数が 2 つの場合; この場合, 方程式はゼロ変数をシステム内に作ることによって別々の係数をもちます. この定式化も H が単位行列 (方程式の誤差間で同じ分散で, 相関がない) であることから幾らか非現実的です.

```
ZERO = 0;
KALMAN(NOET) Y1 Y2 | C1 X1 ZERO ZERO, ZERO ZERO C2 X2;
```

KALMAN

Harvey の例2(p.116-117)(信号+ノイズ,あるいはCooley-Prescott)

```
READ Y; 4.4 4.0 3.5 4.6;
SET Q=4; SET A0=4; SET P0=12;
KALMAN(VT=Q,BPRIOR=A0,VBPRIOR=P0) Y C;
```

Harvey の練習問題1(p.119)(確率収束)

```
READ Y; 4.4 4.0 3.5 4.6;
Y4=Y-4;
SET RH0=.5; SET Q=4; SET A0=.2; SET P0=3;
KALMAN(BT=RH0,VT=Q,BP=A0,VBP=P0) Y4 C;
```

遷移方程式における分散のブートストラップ

```
SMPL 1,100;
KALMAN(NOETRANS) Y C X1 X2;
UNMAKE @STATE B1-B3;
SMPL 4,100;
DOT 1-3;
  D.=B.-B.(-1);
ENDDOT;
COVA D1-D3;
MAT VTOS=@COVA/@S2;
KALMAN(VTRANS=VTOS) Y C X1 X2;
```

MLPROCによるハイパー・パラメータの推定. この例はML PROC を用いて遷移行列 Q の分散を推定します.

```
MFORM(NROW=2,TYPE=SYM) Q;
PARAM Q11,1 Q22,2;
ML KFQ Q11,Q22;
? ML PROCで対数尤度の評価
PROC KFQ;
  IF (Q11 <=0 .OR. Q22 <=0); THEN;      ? 制約のチェック
    SET @LOGL=@MISS;                     ? 分散は > 0か?
  ELSE; DO;
    SET Q(1,1)=Q11; SET Q(2,2)=Q22;
    SUPRES @COEF;
    KALMAN(SILENT,VTRANS=Q) Y C X;
    NOSUPRES @COEF;
  ENDDO;
ENDPROC;
```

アウトプット:

標準的な係数と標準誤差の表が, 最後の状態ベクトルや対数尤度と一緒にプリントされます. 次の統計量が保存されます(プリントも可).

変数	長さ	変数についての内容
@COEF	m	最終の状態ベクトル
@SES	m	標準誤差
@T	m	t 統計量
@VCOV	m × m	分散共分散行列
@LOGL	1	尤度関数の対数
@SSR	1	逐次残差の平方和
@S2	1	逐次残差の分散
@RES1	観測値数 × n	(1 期先) 予測誤差
@RECRES	観測値数 × n	逐次残差 (iid)
@STATE	観測値数 × m	発展的状态ベクトル
@RESD	観測値数 × n	(SMOOTHの時) 直接残差
@SMOOTH	観測値数 × m	平滑化状態ベクトル

残差あるいは状態ベクトルの最初の数行は、それらの観測値が事前分布の計算に使われている場合はゼロとなります。また、OLS 回帰の逐次残差は REGOPT(CALC) RECRES; そして OLS で得ることができます。

方法:

カルマンフィルターは、各観測値に対する y_t と X_t についての新しい情報を用いて β_t の推定値(とその分散)を逐次的に更新しますので、バイズ的な方法であると見ることができます。しかし、利用者は事前分布を与える必要はありません。標本の初期観測値から回帰型のモデルによって自動的に計算されます。実際の更新方程式については Harvey を参照して下さい。(必要な場合に) 平滑化された状態ベクトルは全標本を基に計算されます。これも詳細は Harvey を参照して下さい。

それぞれの時間区間に対する推定方法は、その点までに観測されたデータに関する条件付最尤法です。

σ^2 要素と対数尤度は逐次残差から計算されます。逐次残差は Harvey の記号法に従えば $e_t = \text{CHOL}(F_t)^{-1}v_t$ となりますから、 $E[e_t'e_t] = I$ となります。もし σ^2 を H, Q, P_0 から分解しなければ、推定された @S2 は 1 に近くなるはずで、 $M > 1$ で予測誤差に共線性がある場合は、得られる σ^2 、標準誤差、そして対数尤度に問題が生じます。

参考文献:

Cooley, T.F. and Edward Prescott, "Varying Parameter Regression: A Theory and Some Applications," **Annals of Economic and Social Measurement**, 2 (1973), pp.463-474.

Cooper, J.Philip, "Time-Varying Regression Coefficients: A Mixed Estimation Approach and Operational Limitation of the General Markov Structure," **Annals of Economic and Social Measurement**, 2 (1973), pp.525-530.

Harvey, Andrews C., **Time Series Models**, 1981, Philip Allen, London, pp.101-119.

Harvey, Andrews C., **Forecasting, structural time series models and the Kalman filter**, 1989, Cambridge University Press, New York.

Maddala, G.S., **Econometrics**, 1977, McGraw-Hill Book Co., New York, pp.396-400.